Geometrie v $\mathbb{R}^n$

Začněme nejjednodušší úlohou: Vypočtěme vzdálenost dvou bodů v rovině. Použijeme příkaz **distance** z balíku **student.** Poznamenejme, že vlastně počítáme délku úsečky, která oba body spojuje.

```
> restart;
> with(student):
> distance([a, b], [c, d]);
```

$$\sqrt{(b-d)^2+(c-a)^2}$$

```
> distance([-1, 1], [3, -2]);
```

$$5$$

Obdobně pro body ve vícerozměrném Euklidovském prostoru:

```
> distance([a, b,c,d], [a1,b1,c1,d1]);
```

$$\sqrt{(d1-d)^2+(c1-c)^2+(b1-b)^2+(a1-a)^2}$$

Vypočtěme ještě souřadnice středu úsečky AB, nejprve obecně, pak pro
$A=[-1,1],\ B=[2,-1]:$

```
> midpoint([a, b], [c, d]);
```

$$[\frac{a}{2}+\frac{c}{2}, \frac{b}{2}+\frac{d}{2}]$$

```
> midpoint([-1, 1], [3, -2]);
```

$$[1, \frac{-1}{2}]$$

Vraťme se ještě k příkazu **distance.** Příkaz **student[distance]** počítá vzdálenost dvou bodů v $\mathbb{R}^n$:

```
> with(student):
> distance(-1,1);
```

$$2$$

```
> distance(-1,x+1);
```

$$|x+2|$$

```
> distance([1,2,3,4,5],[0,0,0,0,0]);
```

$$\sqrt{55}$$

Příkaz **geometry[distance]** počítá vzdálenost dvou bodů nebo vzdálenost bodu od přímky:

```
> restart;
> with(geometry):
> point(A,a,b);
```

$$A$$

```
> line(l,x+2*y=1,[x,y]):
> distance(A, l);
```

$$\frac{1}{5}\ |-1+a+2\,b|\ \sqrt{5}$$

```
> a:=1; b:=2;
```

$$a := 1$$
$$b := 2$$

```
> distance(A, l);
```

$$\frac{4\sqrt{5}}{5}$$

Tedy vzdálenost bodu A od dané přímky l je $\frac{4\sqrt{5}}{5}$.

Příkaz **geom3d[distance]** počítá vzdálenost dvou geometrických objektů: bodů, přímek i rovin:

```
> restart;
> with(geom3d):
```

Vzdálenost bodu A od roviny ρ :

```
> plane(rho,x+2*y-z+4=0,[x,y,z]):
> point(A,[a,b,c]):
> distance(A,rho);
```

$$\frac{1}{6}\ |a+2\,b-c+4|\ \sqrt{6}$$

Vzdálenost dvou rovnoběžných rovin:

```
> plane(sigma,x+2*y-z=0,[x,y,z]):  distance(rho,sigma);
```

$$\frac{2\,\sqrt{6}}{3}$$

Znázorněme si nyní dva vektory a vypočtěme úhel, který svírají:

```
> restart;
> with(plots):
> b1 := arrow(<1,2,3>, <3,-4,5>, shape=double_arrow, color=red):
> b2 := arrow(<1,2,3>, <-1,0,2>, shape=double_arrow, color=blue):
> display(b1, b2,axes=FRAMED);
```



Příkaz **angle** balíku **linalg** počítá (v radiánech) úhel alfa dvou $n-$dimenzionálních vektorů **u** a **v** podle vzorce

$$\cos\alpha = \frac{\mathbf{u}.\mathbf{v}}{\|\mathbf{u}\|\ \|\mathbf{v}\|}$$

```
> with(linalg):
> alfa:=angle( vector([2,-6,2]), vector([-2,-2,-1]) );
```

$$alfa := \arccos(\frac{\sqrt{44}\,\sqrt{9}}{66})$$

```
> evalf(%);
```

$$1.264518958$$

```
> convert(alfa, units, rad, deg);
```

$$\frac{180\arccos(\frac{\sqrt{44}\,\sqrt{9}}{66})}{\pi}$$

```
> evalf(%);
```

$$72.45159939$$

Tedy vektory svírají úhel 1,264518958 radiánů, tj. 72,45159936 stupňů.

Tři body určují trojúhelník, který si můžeme pomocí **draw** v balíku **geometry** nakreslit:

```
> with(geometry):
> triangle(T,[point(A,-1,0),point(B,3,-4),point(C,1,2)]);
```

$$T$$

```
> draw([T(color=blue)], printtext=true, axes=normal);
```




```
> IsRightTriangle(T);
```

$$true$$

Trojúhelník T je pravoúhlý. Jaká je plocha T? V balíku **geometry** ji vypočteme příkazem **area:**

```
> area(T);
```

$$8$$

Práce s vektory je podrobně popsána v části Lineární algebra. Na tomto místě připomeňme pouze příkazy **dotprod** (počítá skalární součin dvou vektorů) a **crossprod** (vypočte vektorový součin dvou vektorů z $\mathbb{R}^3$):

```
> with(linalg):with(plots):
> u:=vector([1,1,2]);
```

$$u := [1, 1, 2]$$

```
> v:= vector([-1,-2,1]);
```

$$v := [-1, -2, 1]$$

```
> dotprod(u,v);
```

$$-1$$

```
> crossprod(u,v);
```

$$[5, -3, -1]$$

```
> crossprod(v,u);
```

$$[-5, 3, 1]$$

```
> poc:=vector([0,0,0]);
```

$$poc := [0, 0, 0]$$

```
> b1 := arrow(poc, u, shape=arrow, color=blue):
> b2 := arrow(poc, v, shape=arrow, color=blue):  b3 := arrow(poc,
> crossprod(u,v), shape=arrow, color=red):
> b4 := arrow(poc, crossprod(v,u), shape=arrow, color=green):
> display(b1, b2,b3,b4,axes=FRAMED);
```



Jak z výpočtu, tak z obrázku je patrné, že pro vektory $\mathbf{u} \times \mathbf{v}$ a $\mathbf{v} \times \mathbf{u}$ platí: $\mathbf{v} \times \mathbf{u} = -\mathbf{u} \times \mathbf{v}$.

Příklad: Najděte alespoň jeden vektor **a**, který vyhovuje rovnici $\mathbf{a} \cdot \mathbf{b} = 0$ pro daný vektor **b**.

```
> a := <x | y| z>;
```

$$a := [x, y, z]$$

```
> b:=<-1,4,2>;
```

$$b := \begin{bmatrix} -1 \\ 4 \\ 2 \end{bmatrix}$$

```
> evalm(a&*b);
```

$$-x + 4\,y + 2\,z$$

Všechny vektory, které splňují rovnost $\mathbf{a} \cdot \mathbf{b} = 0$ tvoří rovinu $-x+4y+2z = 0$, vektor $\mathbf{b}$ je její normálový vektor. Rovnici $\mathbf{a} \cdot \mathbf{b} = 0$ splňuje libovolný vektor, který leží v této rovině, např. $\mathbf{a} = (0,0,0)$ nebo $\mathbf{a} = (2,1,-1)$. Jaký je obecný tvar libovolného takového vektoru zjistíme, jestliže řešíme homogenní soustavu lineárních algebraických rovnic $-x+4y+2z = 0$ (1 rovnice pro 3 neznámé, matice soustavy je typu 1x3, vektor neznámých je 3x1, pravá strana je vektor 1x1):

```
> solve({-x+4*y+2*z=0},{x,y,z});
```

$$\{x = 4\,y + 2\,z,\ y = y,\ z = z\}$$

Prostor všech řešení naší homogenní soustavy má tedy dimenzi 2. My obvykle píšeme parametrický tvar roviny ve tvaru $z = t,\ y = s,\ x = 4s + 2t, s, t \in \mathbb{R}$. Připomeňme si, jak lze zjistit vlastnosti naší soustavy pomocí Maplu. Podrobnosti lze najít v části MI, Lineární algebra.

```
> eqns := {-x+4*y+2*z=0};
```

$$eqns := \{-x + 4\,y + 2\,z = 0\}$$

```
> C := genmatrix(eqns, [x,y,z], c);
```

$$C := \begin{bmatrix} -1 & 4 & 2 \end{bmatrix}$$

```
> evalm(c);
```

$$[0]$$

```
> rank(C);
```

$$1$$

```
> nullspace(C,d);
```

$$\{[4,\ 1,\ 0],\ [2,\ 0,\ 1]\}$$

```
> d;
```

$$2$$

Na následujícím obrázku jsme znázornili zadaný vektor $\mathbf{b}$ a množinu (rovinu) všech vektorů, které jsou k němu kolmé.

```
> p1 := arrow([0,0,0], b, shape=arrow, color=red):
> p2:=implicitplot3d({-x+4*y+2*z=0
>},x=-2..2,y=-2..2,z=-2..3):display(p1, p2,axes=FRAMED);
```



Příklad: Najděte alespoň jeden vektor $\mathbf{a}$, který vyhovuje rovnici $\mathbf{a} \cdot \mathbf{b} = 6$ pro daný vektor $\mathbf{b} = (4, 1, -5)$.

Řešení: Tentokrát řešíme nehomogenní soustavu rovnic

```
> solve({4*x+y-5*z=6},{x,y,z});
```

$$\{z = z,\ x = x,\ y = -4\,x + 5\,z + 6\}$$

Vektory, splňující požadovanou rovnost, tvoří opět rovinu, jejíž parametrické rovnice nám Maple napsal: $x = t,\ y = 6 - 4t + 5s,\ z = s,\ t, s \in \mathbb{R}$. V této rovině leží bod $(0, 6, 0)$, směrové vektory jsou vektory $(1, -4, 0)$ a $(0, 5, 1)$.

```
> p1 := arrow(<1,-4,0>, shape=arrow, width = 0.3, thickness=3,
> color=red):
> p2 := arrow(<0,5,1>, shape=arrow,width = 0.3,thickness=3, color=blue):
> p3:=implicitplot3d({4*x+y-5*z=6},x=-5..5,y=-10..10,z=-10..20):
> p4:=arrow(<4,1,-5>, shape=arrow, width=0.3,thickness=3,
> color=black):display(p1, p2,p3,p4,axes=FRAMED);
```



Příklad: Zjistěte, který z bodů $A = (2, 1, 6),\ B = (5, -10, 12),\ C = (14, -2, 0)$ leží v rovině $3x+5y-2z+1 = 0$. Leží zbývající body v jednom poloprostoru určeném danou rovinou?

Řešení: Dosadíme souřadnice bodů A, B, C do rovnice roviny:

```
> ro:=(x,y,z)->3*x+5*y-2*z+1;
```

$$ro := (x,\ y,\ z) \to 3\,x + 5\,y - 2\,z + 1$$

```
> ro(2,1,6);
```

$$0$$

Bod A tedy leží v dané rovině.

```
> ro(5,-10,12);
```

$$-58$$

```
> ro(14,-2,0);
```

$$33$$

Body B a C neleží v dané rovině. Leží v různých poloprostorech tvořených touto rovinou. Znázorněme si situaci ještě graficky:

```
> with(plottools):
> a := point([2,1,6], color=red, symbol=diamond):
> b:=point([5,-10,12], color=red, symbol=diamond):  c:=point([14,-2,0],
> color=red, symbol=diamond):
> p3:=implicitplot3d({3*x+5*y-2*z+1=0},x=-2..20,y=-5..5,z=-3..20):
> display(a,b,c,p3,orientation=[60,60],axes=normal);
```



Příklad: Určete vzájemnou polohu přímek AB a CD, je-li $A = (2, -5, -2),\ B = (0, -3, 0),\ C = (4, 1, 2),\ D = (-1, -2, 1)$.

Řešení: Připomeňme, že dvě přímky jsou rovnoběžné, jsou-li jejich směrové vektory lineárně závislé (tj. v případě dvou vektorů je jeden násobkem druhého). Jsou-li směrové vektory lineárně nezávislé a mají-li přímky jeden společný bod, jsou různoběžné. Nemají-li společný bod, jsou mimoběžné.
Postupujeme tedy takto: najdeme směrové vektory přímek AB a CD, zjistíme, zda jsou tyto vektory lineárně závislé nebo nezávislé. Jsou-li nezávislé, hledáme společný bod přímek.

```
> with(geometry):
> u:=vector(3,[-2,2,2]);
```

$$u := [-2,\ 2,\ 2]$$

```
> v:=vector(3,[-5,-3,-1]);
```

$$v := [-5,\ -3,\ -1]$$

Tedy vektory **u** a **v** jsou lineárně nezávislé, přímky nejsou rovnoběžné, mohou být různoběžné nebo mimoběžné. Sestavíme parametrické rovnice obou přímek a porovnáme příslušné souřadnice eventuálního průsečíku. Parametrické rovnice:
$AB \equiv\ x = 2-2t,\ y = -5+2t,\ z = -2+2t,\ t \in \mathbb{R}$
$CD \equiv\ x = 4-5s,\ y = 1-3s,\ z = 2-s,\ s \in \mathbb{R}.$
Řešíme soustavu tří rovnic pro dvě neznámé t a s: $2-2t = 4-5s,\ \ -5+2t = 1-3s,\ \ -2+2t = 2-s$:

```
> eqns := {2-2*t=4-5*s, -5+2*t=1-3*s, -2+2*t=2-s};
```

$$\mathit{eqns} := \{2-2\,t = 4-5\,s,\ -5+2\,t = 1-3\,s,\ -2+2\,t = 2-s\}$$

```
> C := genmatrix(eqns, [t,s], 'b');
```

$$C := \begin{bmatrix} -2 & 5 \\ 2 & 3 \\ 2 & 1 \end{bmatrix}$$

```
> print(b);
```

$$[2,\ 6,\ 4]$$

```
> eqns1:=geneqns(C,[t,s],b);
```

$$\mathit{eqns1} := \{-2\,t+5\,s = 2,\ 2\,t+3\,s = 6,\ 2\,t+s = 4\}$$

```
> solve(eqns1);
```

$$\{s = 1,\ t = \frac{3}{2}\}$$

Soustava má právě jedno řešení, přímky jsou tedy různoběžky a protínají se v bodě $P = (P1, P2, P3)$, jehož souřadnice dostaneme buď dosazením hodnoty $t = 3/2$ do parametrické rovnice přímky AB nebo dosazením hodnoty $s = 1$ do rovnice přímky CD :

```
> t:=3/2;
```

$$t := \frac{3}{2}$$

```
> P1:=2-2*t; P2=-5+2*t; P3=-2+2*t;
```

$$\mathit{P1} := -1$$
$$\mathit{P2} = -2$$
$$\mathit{P3} = 1$$

```
> s:=1;
```

$$s := 1$$

```
> P:=(4-5*s, 1-3*s, 2-s);
```

$$P := -1,\ -2,\ 1$$

Příklad: Napište rovnici roviny, znáte-li patu kolmice $P = (4, -1, 3)$ spuštěné z počátku na tuto rovinu.

Řešení: Sestavíme obecný tvar rovnice roviny: $ax+bz+cz+d = 0$, kde a, b, c jsou složky normálového vektoru dané roviny. Normálový vektor je v tomto případě zadán jako vektor OP, kde $O = (0, 0, 0)$ je počátek. Protože bod P leží v hledané rovině, musí rovnici splňovat. To využijeme k určení d.

```
> OP:=vector(3,[4,-1,3]);
```

$$\mathit{OP} := [4,\ -1,\ 3]$$

```
> a:=4; b:=-1; c:=3;
```

$$a := 4$$
$$b := -1$$
$$c := 3$$

```
> rho:=(x,y,z)->a*x+b*y+c*z+d;
```

$$\rho := (x,\, y,\, z) \to a\,x + b\,y + c\,z + d$$

```
> rho(x,y,z);
```

$$4\,x - y + 3\,z + d$$

```
> solve(rho(4,-1,3)=0,{d});
```

$$\{d = -26\}$$

```
> d:=-26;
```

$$d := -26$$

```
> rho(x,y,z);
```

$$4\,x - y + 3\,z - 26$$

Toto je hledaná rovnice roviny.

Příklad: Určete vzájemnou polohu přímky p a roviny ρ. V případě, že jsou různoběžné, najděte jejich průsečík a úhel, který svírají:

$$p \equiv \;\; x = 2 + t,\;\; y = 1 - 3t,\;\; z = 5t,\;\; t \in \mathbb{R}; \quad \rho \equiv \;\; 2x - y - z = 3.$$

Řešení: Nejprve určíme směrový vektor $\mathbf{u}$ přímky p a normálový vektor $\mathbf{n}$ roviny ρ :

```
> u:=vector(3,[1,-3,5]);
```

$$u := [1,\, -3,\, 5]$$

```
> n:=vector(3,[2,-1,-1]);
```

$$n := [2,\, -1,\, -1]$$

```
> dotprod(u,n);
```

$$0$$

Vektory $\mathbf{u}$ a $\mathbf{n}$ jsou navzájem kolmé, tedy přímka je rovnoběžná s rovinou. Ověříme ještě, zda přímka p neleží v rovině ρ. Stačí zjistit, zda jeden bod přímky, např. $(2, 1, 0)$, leží v rovině ρ.

```
> x:=2; y:=1; z:=0;
```

$$x := 2$$
$$y := 1$$
$$z := 0$$

```
> eval(2*x-y-z-3);
```

$$0$$

Bod splňuje rovnici roviny a přímka p tedy leží v rovině ρ.

Příklad: Napište obecnou i parametrickou rovnici roviny ρ, která prochází bodem $M = (-2, 7, 3)$ a je rovnoběžná s rovinou σ určenou body $A = (1, 0, 0),\; B = (0, 1, 1),\; C = (-6, 2, 3)$.

Řešení: Obě roviny mají stejné směrové vektory AB a AC a bod M leží v rovině ρ.

```
> restart;
> with(plots):  with(linalg):
> A:=(1,0,0); B:=(0,1,1); C:=(-6,2,3);
```

$$A := 1,\, 0,\, 0$$
$$B := 0,\, 1,\, 1$$
$$C := -6,\, 2,\, 3$$

```
> AB:=vector(3,[B-A]);
```

$$AB := [-1,\, 1,\, 1]$$

```
> AC:=vector(3,[C-A]);
```

$$AC := [-7,\, 2,\, 3]$$

Výsledná parametrická rovnice roviny ρ je tedy: $x = -2 - t - 7s,\; y = 7 + t + 2s,\; z = 3 + t + 3s,\; s, t \in \mathbb{R}$. Vypočteme vektorový součin vektorů AB a AC a získáme tak normálový vektor obou rovin.

```
> n:=crossprod(AB,AC);
```

$$n := [1,\, -4,\, 5]$$

Obecná rovnice roviny $\rho \equiv\ x - 4y + 5z + d = 0$, zbývá zjistit $d:$

```
> solve(-2-4*7+5*3+d=0,d);
```

$$15$$

Tedy $\rho \equiv\ x - 4y + 5z + 15 = 0$.

Příklad: Najděte společné body rovin: $x + y + z = 1,\ y + z = -1,\ x + y - z = 2$.

Řešení: Body, které leží ve všech třech rovinách, musí splňovat všechny tři rovnice. Hledáme tedy řešení soustavy tří rovnic pro tři neznámé.

```
> solve({x+y+z=1,y+z=-1,x+y-z=2},{x,y,z});
```

$$\{x = 2,\ y = \frac{-1}{2},\ z = \frac{-1}{2}\}$$

```
> implicitplot3d({x+y+z=1,y+z=-1,x+y-z=2},x=-3..3,y=-3..3,z=-3..3,
> axes=normal);
```



Všechny tři roviny mají jeden společný bod.

Příklad: Najděte společné body rovin: $x + y + z = 1,\ x - y + 2z = 1,\ 2x + 3z = 2$.

Řešení:

```
> solve({x+y+z=1,x-y+2*z=1,2*x+3*z=2},{x,y,z});
```

$$\{x = 1 - 3\,y,\ z = 2\,y,\ y = y\}$$

```
> implicitplot3d({x+y+z=1,x-y+2*z=1,2*x+3*z=2
>},x=-3..3,y=-3..3,z=-3..3);
```



Soustava má nekonečně mnoho řešení, průsečnice těchto tří rovin je přímka.

Příklady k procvičení

1. Napište obecnou rovnici roviny, ve které leží bod $A = (0; -1; 0)$ a která je rovnoběžná s přímkami

$$\begin{array}{lllll} p: & x = 2 + 2t\,, & y = t\,, & z = 0\,, & t \in \mathcal{R}\,, \\ q: & x = -3 - 3s\,, & y = -2 - s\,, & z = 2 + 2s\,, & s \in \mathcal{R}\,. \end{array}$$

Leží některá z daných přímek v této rovině?

2. Zjistěte, zda bod $A = (-3; 4; 1)$ leží v rovině dané rovnicemi

$$x = -1 + \alpha + 4\beta\,, \qquad y = -3\alpha - 2\beta\,, \qquad z = 3 - \alpha + \beta\,,$$

a napište rovnici přímky, která tímto bodem prochází a je k dané rovině kolmá.

3. Vypočtěte úhel přímky

$$p: \; x = -8 - 2t\,, \quad y = 8 + 3t\,, \quad z = 3 + t\,, \quad t \in \mathcal{R}\,,$$

a roviny

$$\rho: \; x = -2 - 2\beta\,, \quad y = -3 - \alpha + 3\beta\,, \quad z = 2 + \alpha + \beta\,, \quad \alpha,\ \beta \in \mathcal{R}\,,$$

a nalezněte jejich společné body.

4. Vypočtěte úhel rovin

$$\begin{array}{ll} \rho: & x = 1 + \alpha - 3\beta\,, \quad y = 1 + 2\alpha\,, \quad z = 1 + 2\alpha + 4\beta\,, \quad \alpha,\ \beta \in \mathcal{R}\,, \\ \sigma: & 2x - 2y - z + 16 = 0 \end{array}$$

a napište parametrické rovnice přímky, která je k rovině ρ kolmá a prochází bodem $A = [1, 1, 1]$.

5. Určete reálná čísla a, b tak, aby nadroviny v prostoru $\mathbb{E}^4$ dané rovnicemi

$$2x_1 + 2x_2 + 2x_3 - x_4 = -2\,, \quad x_1 + 3x_3 + 2x_4 = 2\,, \quad 2x_2 + ax_3 - 5x_4 = b$$

měly společnou rovinu. Napište parametrické rovnice této roviny.

6. Najděte společné body nadrovin

$$\begin{array}{llllll} \rho: & x_1 = 2 - \alpha\,, & x_2 = 1 - \alpha + \beta\,, & x_3 = \alpha + 2\beta\,, & x_4 = 2\alpha - \beta\,, & \alpha\,, \beta \in \mathcal{R}\,, \\ \sigma: & x_1 = 2 + 2\gamma + \delta\,, & x_2 = 2 - 2\gamma + \delta\,, & x_3 = -5 - \delta\,, & x_4 = -1 + \gamma - 2\delta\,, & \gamma\,, \delta \in \mathcal{R}\,. \end{array}$$

Jaký geometrický útvar tyto body vyplní? Napište jeho parametrické rovnice.

7. Napište parametrické rovnice roviny, ve které leží bod $A = (-1; 2; 0; 2)$ a která je rovnoběžná s přímkami

$$\begin{array}{llllll} p: & x_1 = -1 - t\,, & x_2 = 2 - 2t\,, & x_3 = t\,, & x_4 = 2\,, & t \in \mathcal{R}\,, \\ q: & x_1 = s\,, & x_2 = 3 + 3s\,, & x_3 = 3 - s\,, & x_4 = 2 - s\,, & s \in \mathcal{R}\,. \end{array}$$

Leží některá z daných přímek v této rovině?